

Appendix A. Supplemental Infoprint Server information

This appendix contains additional information needed to develop an e-mail exit program. It also contains detailed field listings for the data queue used by the PDF subsystem.

A.1 E-mail exit program

The following is a template for an e-mail exit program using a combination of RPGLE and CL. This program is referenced in the End to End Case Study chapter and the Infoprint Server PDF Creation chapter.

A sample C Language program can be found in the *Infoprint Server for iSeries User's Guide*, G544-5775.

A.1.1 RPGLE e-mail exit program mainline

Use this sample RPGLE program as a template to customize your own e-mail application.

```

FLOOKUP  IF  E          K DISK
D*
D INPUTS          DS
D JOBNAM          1      26
D SPLFID          27     36
D SPLNO           37     40B 0
D MAILTAG         41     290
D PDFFILE         291    630
D RES1            631    632
D PATHCCSID       633    636B 0
D SENDER          637    646
D USRDTA          647    656
D*****
D OUTDS          DS
D DISPOSTN       1      1
D RES2           2      4
D MSGLEN         5      8B 0
D ADDRLEN        9     12B 0
D MSGTEXT        13    267
D RES3           268    287
D ADDRESS        288    542
D*
C      *ENTRY      PLIST
C              PARM          INPUTS          656
C              PARM          INPUTLEN        4 0
C              PARM          OUTDS           542
C              PARM          OUTPUTLEN       4 0
C              PARM          OUTINFO         4 0
C*
C              EVAL          MSGLEN         = 255
C              EVAL          ADDRLEN        = 255
C              EVAL          OUTPUTLEN      = 542
C              EVAL          OUTINFO        = 542
C*
C              MOVE          *BLANKS        ADDRESS
C              SUBST         MAILTAG        CUSTNO          6
C              CUSTNO       CHAIN          LOOKUP
C              IF           NOT %FOUND
C              EXSR         Not_Found
C              RETURN
C              ENDIF
C*
C              MOVE         EADDR          ADDRESS
C*
C              EVAL          MSGTEXT = 'Hello ` + %TRIMR(PERSON)
C              + `, this is your invoice for `
C              + %TRIMR(COMP) + `.`
C              EVAL          DISPOSTN = '1'
C*

```

```

C          RETURN
C*****
C      Not_Found      BEGSR
C*****
C
C          CALL      'NOTFOUND'
C          PARM          CUSTNO
C          PARM          PDFFILE
C          EVAL      DISPOSTN = '0'
C          ENDSR
C*
    
```

A.1.2 NOTFOUND program called by EMAIL

Sample program called by e-mail exit program when lookup operation fails. You may add other commands to this program to perform other functions such as change authority of the generated PDF file in the IFS, or to rename it.

```

PGM      PARM(&CUSTNO &PDFFILE)
DCL      VAR(&CUSTNO) TYPE(*CHAR) LEN(6 )
DCL      VAR(&PDFFILE) TYPE(*CHAR) LEN(340)

SNDMSG   MSG('Cannot find e-mail address for customer +
            number' *BCAT &CUSTNO *BCAT '. File' +
            *BCAT &PDFFILE *BCAT 'not sent.') +
            TOUSR(MSHNIE) MSGTYPE(*INFO)
CPY      OBJ(&PDFFILE) TODIR('/mira')
MONMSG   MSGID(CPFA09C CPFA0A0)

ENDPGM
    
```

A.1.3 Sample DDS for LOOKUP file

This is a sample DDS layout for the lookup file used by the e-mail exit program.

```

A          UNIQUE
A          R EMAILR
A          CUSTNO      6A      COLHDG('CUSTOMER' 'NUMBER')
A          COMP      25A      COLHDG('COMPANY' 'NAME')
A          PERSON    25A      COLHDG('CONTACT')
A          EADDR     80A      COLHDG('E-MAIL' 'ADDRESS')
A          K CUSTNO
    
```

A.1.4 Exit program parameters

The e-mail exit program uses the following parameters.

Table 1. E-mail exit program parameters

Description	Usage	Type	Comments
E-mail exit input information	INPUT	CHAR(*)	The e-mail exit program information which is input to the e-mail exit program from the PSF/400 print writer.
Length of input information	INPUT	BINARY(4)	The length of the input information
E-mail exit output information	OUTPUT	CHAR(*)	The e-mail exit program information which is output from the e-mail exit program to the PSF/400 print writer.

Description	Usage	Type	Comments
Length of output information buffer	INPUT	BINARY(4)	The size, in bytes, of the mail tag output information buffer.
Length of output information available	OUTPUT	BINARY(4)	Length of the output information available. To avoid losing information this should be less than the e-mail tag output information.

A.1.5 E-mail exit program input data structure

The following table shows the structure for the e-mail input parameter.

Table 2. E-mail input information

Position	Type	Description	Comments
1	CHAR(26)	Qualified job name	
27	CHAR(10)	Spooled file name	
37	BINARY(4)	Spooled file number	
41	CHAR(250)	Mail tag	As provided by DDS STRPAGGRP or generated by CRTAFPDTA
291	CHAT(340)	Path and name of PDF file	
631	CHAR(2)	Reserved	
633	BINARY(4)	Path name CCSID	
637	CHAR(10)	Mail sender	
647	CHAR(10)	User Data (USRDTA)	

A.1.6 E-mail exit program output data structure

Table 3. E-mail output information

Position	Type	Description	Comment
1	CHAR(1)	Disposition of the PDF file	0=do not mail 1=e-mail to specified address
2	CHAR(3)	Reserved	Set to null
5	BINARY(4)	Length of message text	0 to 255 bytes

Position	Type	Description	Comment
9	BINARY(4)	Length of mail address data	0-16,000,000
13	CHAR(255)	Message text data	
268	CHAR(20)	Reserved	initialized to null
288	CHAR(*)	E-mail address(es)	Each address must be delimited by single quotation marks.

A.2 PDF data queue structure

The following chart describes the field layout for the PDF Data queue. An entry is made in this data queue (if specified in the PSF Configuration object) for every PDF file created. Information on specifying and using this queue can be found in the End to End Case Study chapter and the Infoprint Server PDF Creation chapter.

Table 4. PDF transform data queue parameters

From Position	To Position	Data Type	Description
1	10	CHAR(10)	Function: Identifies the function that created the data queue entry. The value for converting a spooled file to PDF is *PDFWTR.
11	12	CHAR(2)	Record type: Identifies the record type within the function. For PDF conversion the value is '01'
13	14	CHAR(2)	Return value: 01 - File successfully converted and stored in specific folder. 02 - File successfully converted and placed on specified output queue. 03 - File successfully converted and electronically mailed. 04 - File successfully converted, but e-mail exit program specified not to mail PDF file. 05 - File successfully converted, but not mailed due to invalid mail tag or address. 06 - Conversion error. 07 - File could not be spooled to the output queue.
15	40	CHAR(26)	Qualified Job Name: CHAR(10) - Job name CHAR(10) - User name CHAR(6) - Job number

From Position	To Position	Data Type	Description
41	50	CHAR(10)	Spoiled file name: Identifies the name of the spoiled file that was converted to PDF.
51	52	CHAR(2)	Reserved
53	56	BINARY(4)	Spoiled file number: Identifies the unique number of the spoiled file that was converted to PDF.
57	306	CHAR(250)	Mail tag: Identifies the mail address or mail tag specified on the printer file USRDFNDA parameter or the DDS STRPAGGRP keyword.
307	646	CHAR(340)	Path and name of the PDF file: Identifies the path and name of the PDF file.
647	648	CHAR(2)	Reserved
649	652	BINARY(4)	Path name CCSID
653	662	CHAR(10)	Mail sender: Identifies the sender of the electronic mail. The value is specified on the printer file USRDFNDA parameter or on the PSF configuration object PDFSENDER parameter.
663	672	CHAR(10)	User Data
673	752	CHAR(80)	Reserved

